Équations

Table des matières

1 Systèmes d'équations linéaires **1**

2 Équations non linéaires **1**

3 Équations algébriques **1**

4 Forcer l'évaluation numérique **2**

5 Inégalités, inéquations et système d'inéquations **2**

6 Résolution d'équations différentielles **3**

7 Résolution d'une relation de récurrence **3**

---

1 Systèmes d'équations linéaires

Rappel : un système de n équations linéaires à n inconnues possède une et une seule solution ssi le déterminant du système est non nul. Le programme de la classe de PCSI ne comporte que le calcul du déterminant d'une matrice carrée d'ordre 2 ou 3.
La syntaxe requise pour la résolution d'un système d'équations (linéaires ou non) est :

```
solve({eq_1, ... ,eq_n},{var_1, ... ,var_p}) ;
```

où $eq_1, \ldots, eq_n$ sont les équations et $var_1, \ldots, var_p$ sont les inconnues. Voici un exemple :

```
solve({2*x+3*y=22,3*x-y=11},{x,y}) ;
```

Maple répond `{x=5,y=3}`.
Exercice : avec Maple, résolvez le système constitué des trois équations $x + 2y + 3z = 1$, $2x + 3y + z = 5$ et $3x + y + 2z = 9$.

2 Équations non linéaires

Voyons ce qui se passe lorsque l'une des équations n'est pas linéaire :

```
solve({x**2+3*y**2=61,2*x+y=16},{x,y}) ;
```

Maple trouve deux solutions : $x = 7$, $y = 2$ et $x = 101/13$, $y = 6/13$.
Selon la complexité de l'équation, Maple rendra ou ne rendra pas une réponse. Voici deux exercices simples.
Exercice : résolvez le système constitué des deux équations $x^2 + y^2 = 169$ et $\ln(x) + \ln(y) = \ln(60)$.
Exercice : résolvez l'équation $\ln\bigl(|x-1|\bigr) + \ln\bigl(|x+2|\bigr) = \ln\bigl(|4x^2+3x-7|\bigr)$

3 Équations algébriques

Une équation algébrique de degré n s'écrit $\sum\limits_{0 \leqslant k \leqslant n} a_k x^k = 0$. La résolution peut s'effectuer dans $\mathbb{R}$ ou dans $\mathbb{C}$. Les équations algébriquess du second degré ne posent pas de problème particulier. Maple sait également résoudre les équations algébriques de degré 3 ou 4 :

```
solve(x**3-3*x**2+4*x-2) ;
```

Maple trouve la solution évidente 1, et les solutions complexes $1+i$ et $1-i$. Observez la syntaxe abrégée : comme il n'y a qu'une équation et une inconnue, les accolades sont inutiles et il n'est pas nécessaire de préciser l'inconnue !

En général, les solutions ont une forme compliquée ; voyons ceci sur un exemple :

```
solve(x**3-x-2) ;
```

Constatez que Maple nous propose une réponse n'apportant aucune information :

```
sols := RootOf(_Z^5-3_Z^2+1,index=1), RootOf(_Z^5-3_Z^2+1,index=2),
RootOf(_Z^5-3_Z^2+1,index=3),RootOf(_Z^5-3_Z^2+1,index=4),
RootOf(_Z^5-3_Z^2+1,index=5) ;
```

Ceci dit, la fonction `sum` permet de calculer certaines fonctions symétriques des racines. Voyons ceci sur un exemple : la commande `sum(x^10,x=RootOf(X^4+p*X+q,X)) ;` nous donne $-10p^2q$. La commande suivante applique `seq` pour calculer d'un coup plusieurs fonctions symétriques des racines.

```
seq(sum(x^k,x=RootOf(x^4+p*x^3+q*x^2+r*x+s,x)),k=1..4) ;
```

Ceci montre qu'il n'est pas difficile de se constituer un jeu d'exercices sur ce sujet, avec des questions de niveaux très variés ; et le tout, sans avoir besoin de faire les calculs à la main !

4 Forcer l'évaluation numérique

Considérons l'équation suivante :

```
sols := solve(x**5-3*x**2+1) :
```

Maple répond`RootOf(x^5-3x^2+1)`, ce qui n'est pas très parlant. Nous pouvons forcer l'évaluation numérique comme suit :

```
allvalues(sols) ;
```

5 Inégalités, inéquations et système d'inéquations

On peut demander à Maple de résoudre une inégalité :

```
solve(3*x+1>0) ;
solve(3*x+1>0,{x}) ;
solve({3*x+1>0,2*x-5<0},{x}) ;
```

Les réponses successives sont $RealRange(Open(-\frac{1}{3}),\infty)$, $\{-\frac{1}{3}<x\}$ et $\{-\frac{1}{3}<x, x<\frac{5}{2}\}$.

Attention ! Ceci ne fonctionne qu'avec des fonctions affines !

Le paquetage `plots` permet de visualiser la résolution d'un système d'inéquations ; voici un exemple :

```
inequal({y<3*x+4,y>-1,y+2*x<=7},x=-4..5,y=-10..10,
 optionsexcluded=(color=white)) ;
```

Exercice : utilisez les idées qui viennent d'être présentées, pour visualiser le triangle de sommets $A(1)$, $B(j)$ et $C(j^2)$ où $j=\frac{-1+\sqrt{3}}{2}$.

Exercice : en utilisant des inégalités bien choisies, représentez un octogone régulier, centré à l'origine, et inscrit dans un cercle de rayon 4. Observez que cet énoncé est imprécis !

6 Résolution d'équations différentielles

La syntaxe à utiliser est la suivante :

```
dsolve({eq_diff,cond_ini},{f_1(var),f_2(var), ... ,f_n(var)}) ;
```

où `eq_diff` est l'équation différentielle (ou la séquence d'équations différentielles), `cond_ini` est la condition initiale (ou la séquence de séquence initiales) et les `f_i(var)` sont les fonctions inconnues. Voici un exemple simple :

```
eq := 2*diff(y(x),x)+y(x)=cos(x) ;
y0 := y(x) ; y1 := diff(y0,x) ;
eq := 2*y1+y0=cos(x) ;
dsolve(eq,y(x)) ;
```

Nous obtenons ainsi l'*équation* d'une courbe intégrale de l'équation différentielle proposée. Si nous voulons une *solution*, il faut remplacer la dernière ligne par ceci :

```
sol := dsolve(eq,y(x)) ; subs(sol,y(x)) ;
```

Pour visualiser une famille de solutions, procéder comme suit, sachant que `lhs` (resp. `rhs`) rend le membre de gauche (resp. de droite) de l'équation considérée.

```
eq := diff(y(x),x)+2*x*y(x)-exp(x-x**2) = 0 ;
s := dsolve(eq,y(x)) ;
S := seq(rhs(dsolve({eq,y(0)=i},y(x))),i=-5..5) ;
plot([S],x=-5..5) ;
```

7 Résolution d'une relation de récurrence

Nous avons vu en cours plusieurs exemples de suites de réels définies par une relation de récurrence du premier ou du deuxième ordre. Maple peut vous aider à résoudre de telles équations.
Prenon un exemple classique, la suite de FIBOACCI. Elle est définie par la donnée de $F_0 = 0$, $F_1 = 1$ et la relation de récurrence $F_n = F_{n-1} + F_{n-2}$ pour $n \geqslant 2$. Nous allons forcer Maple à nous donner une expression simple de F_n. Pour ce faire, nous utilisons `rsolve` :

```
solfib := rsolve({F(n)=F(n-1)+F(n-2),F(0)=0,F(1)=1},{F}) ;
rhs(op(1,solfib)) ;
```

La deuxième commande nous donne l'expression de F_n. Quel équivalent simple de F_n déduisez-vous du résultat proposé par Maple ?

Exercice : utilisez `rsolve` pour obtenir une expression simple du terme général de la suite $(x_n)_{n\in\mathbb{N}}$ définie par $x_0 = 1$ et la relation de récurrence $x_{n+1} = 3x_n + 2^{n+2} - 1$.

FIN